
Change 1 

	Function name
	fl_TC_10_3_1_TestProcedure

	Reason for change
	As per current TTCN implementation the test case may fail with a MultiPDN UE during sub test 1 (IPv4) because during the additional PDN connection the value of “vc_EUTRA_Global.MobileInfo.AssignedPdnType” is set back to the PDN type as supported by the UE (this happens when f_EUTRA_GetPDNAddress() is called during a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step1_2()). Following this an IPv4 address may not be configured during dedicated bearer establishment, causing the UE to loop back packets on the default bearer, contrary to the test case expectation.

	Summary of change
	Following the registration procedures, set the value of “vc_EUTRA_Global.MobileInfo.AssignedPdnType” according to v_PDN_TypeToBeUsed.

	TTCN module
	ESM_Testcases.ttcn

	MCC160 Comment
	Accepted in principle.

Considering that

- TC_10_9_1 has the same issue
- In general when there is an additional PDN its AssignedPdnType can be different to the first one
- So far there has been no parameter of PDN_TypeToBeUsed_Type for the establishment of an additional PDN
- The first PDN does not need to be the one which is kept for the test case body

the following changes were done

- a new global variable PDN_TypeToBeUsed of type PDN_TypeToBeUsed_Type being set to pdnTypeAsSupportedByUE per default was introduced- the variable is set to forceIPv4only or forceIPv6only in TC_10_3_1 and TC_10_9_1
- AssignedPdnType was changed to be an array
- parameter p_PDN_TypeToBeUsed was removed from all the initial registration and preamble functions

MCC160 Implementation see below.


Before change

	    function fl_TC_10_3_1_TestProcedure(boolean p_IpTypeIsIPv4) runs on EUTRA_PTC

  {

    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed;

    if (p_IpTypeIsIPv4) { v_PDN_TypeToBeUsed := forceIPv4only; }   // @sic R5-120541 sic@

    else                { v_PDN_TypeToBeUsed := forceIPv6only; }   // @sic R5-120541 sic@

    //@siclog "Step 0" siclog@

    // The SS performs the generic procedure to get UE in Loopback Activated (state 4)

    // with a Reference default EPS bearer context #1

    // and a Reference dedicated EPS bearer context #1 according to TS 36.508 on Cell A

    // Now continue with the procedure

    f_EUTRA_InitialRegistration(eutra_CellA, PREAMBLE, STATE2A_TESTLOOP_ModeB, NORMAL, -, -, v_PDN_TypeToBeUsed); // @sic R5-160947 sic@

    // UE is in Loopback Activated(state 4) according to TS 36.508 on Cell A.

    f_EUTRA_RbEst_508_AM ( eutra_CellA );

    f_EUTRA_CloseUE_TestLoopModeB ( eutra_CellA, '00'O );

    // steps 1-5

    fl_TC_10_3_1_ProcedureSteps1to5(p_IpTypeIsIPv4);

  } // fl_TC_10_3_1_TestProcedure



After change

	  function fl_TC_10_3_1_TestProcedure(boolean p_IpTypeIsIPv4) runs on EUTRA_PTC

  {

    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed;

    var B3_Type v_PdnTypeValue;
    if (p_IpTypeIsIPv4) { v_PDN_TypeToBeUsed := forceIPv4only; }   // @sic R5-120541 sic@

    else                { v_PDN_TypeToBeUsed := forceIPv6only; }   // @sic R5-120541 sic@

    //@siclog "Step 0" siclog@

    // The SS performs the generic procedure to get UE in Loopback Activated (state 4)

    // with a Reference default EPS bearer context #1

    // and a Reference dedicated EPS bearer context #1 according to TS 36.508 on Cell A

    // Now continue with the procedure

    f_EUTRA_InitialRegistration(eutra_CellA, PREAMBLE, STATE2A_TESTLOOP_ModeB, NORMAL, -, -, v_PDN_TypeToBeUsed); // @sic R5-160947 sic@

select (v_PDN_TypeToBeUsed) {

      case (forceIPv4only) {

        v_PdnTypeValue := tsc_PdnType_IPv4;

      }

      case (forceIPv6only) {

        v_PdnTypeValue := tsc_PdnType_IPv6;

      }

    }
    f_EUTRA_MobileInfo_SetAssignedPdnType(v_PdnTypeValue);
    // UE is in Loopback Activated(state 4) according to TS 36.508 on Cell A.

    f_EUTRA_RbEst_508_AM ( eutra_CellA );

    f_EUTRA_CloseUE_TestLoopModeB ( eutra_CellA, '00'O );

    // steps 1-5

    fl_TC_10_3_1_ProcedureSteps1to5(p_IpTypeIsIPv4);

  } // fl_TC_10_3_1_TestProcedure 


MCC160 Implementation
new global variable PDN_TypeToBeUsed; AssignedPdnType for both PDNs
	  type record EUTRA_MobileInfo_Type {           /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS, UTRAN) */

    NAS_Tmsi    Tmsi,

    NAS_P_Tmsi  P_Tmsi,

    UE_EUTRA_Capability EUTRA_Capability, 
    EUTRA_PdnInfo_Type PdnInfo,                 /* @sic R5s170307 sic@ */


    EUTRA_FDD_TDD_Mode_Type FDD_TDD_Info optional,        // Contain the FDD/TDD mode of the EUTRA cell the UE is currently in @sic 5s150293 sic@

    EmergencyNumList    LocalEmergencyNumberList optional,  /* emergency number list sent by UE in the ATTACH in the preamble @sic R5-125276 sic@ @sic R5s141315 sic@ */

    CoOrd_OtherRATState_Type OtherRATState  optional        // @sic R5-163054 sic@

  };

	  type record EUTRA_PdnInfo_Type {   

    PDN_TypeToBeUsed_Type PdnTypeToBeUsed,

    B3_Type AssignedPdnType[2]

  };

	  function fl_EUTRA_Common_Init(InitParams_Local_Type p_Params) runs on EUTRA_PTC

  {

    vc_EUTRA_Global.Ctrl.Default := null;

    vc_EUTRA_Global.Ctrl.TestBody := false;

    vc_EUTRA_Global.Ctrl.VerdictCtrl := normal;

    vc_EUTRA_Global.Ctrl.HarqErrorHandling := ignore;     /* @sic R5-106675 change 1 sic@ */

    vc_EUTRA_Global.Ctrl.GrantType := USE_DEF_GRANTS;     /* @sic R5s140676 Change 2 sic@ */

    vc_EUTRA_Global.Ctrl.HandleUlDataUM := omit;          /* @sic R5s150338 - dynamic DRB configuration sic@ */

    vc_EUTRA_Global.Ctrl.EUTRA_4_5A_16_State := step1_2;  /* @sic R5-145787 - Implementation for C2K sic@ */

    vc_EUTRA_Global.Ctrl.LoopbackIsActivated := false;  /* @sic R5-153971 sic@ */

    // Initialise security parameters

    vc_EUTRA_Global.Security := f_EUTRA_Security_Init();

    // Initialise Cell array

    if (ischosen(p_Params.NonNas)) {

      vc_EUTRA_Global.CellArray := f_EUTRA_CellArray_Init(p_Params.NonNas.SysinfoCombination);

    } else {

      vc_EUTRA_Global.CellArray := f_EUTRA_NAS_CellArray_Init(p_Params.Nas.CellsOnSamePLMN, p_Params.Nas.SysinfoCombination, p_Params.Nas.CellFreq);

    }

    //Initialise Mobile Info

    vc_EUTRA_Global.MobileInfo.Tmsi := '00000000'O;

    vc_EUTRA_Global.MobileInfo.P_Tmsi := '00000000'O; 
    f_EUTRA_MobileInfo_SetAssignedPdnType(PDN_1);

    f_EUTRA_MobileInfo_SetAssignedPdnType(PDN_2);

    f_EUTRA_MobileInfo_SetPdnTypeToBeUsed(pdnTypeAsSupportedByUE);

    vc_EUTRA_Global.MobileInfo.LocalEmergencyNumberList := omit; /* @sic R5s141315 sic@ */

    vc_EUTRA_Global.MobileInfo.OtherRATState := { /* @sic R5-163054 sic@ */

      CSInOtherRAT := false,

      IMS_Registered_WithExtraPDP := false

    };

    f_EUTRA_ActivateDefault(StandardDefault);

  }


f_EUTRA_MobileInfo_SetPdnTypeToBeUsed
	  function f_EUTRA_MobileInfo_SetPdnTypeToBeUsed(PDN_TypeToBeUsed_Type p_PdnTypeToBeUsed) runs on EUTRA_PTC

  {

    vc_EUTRA_Global.MobileInfo.PdnInfo.PdnTypeToBeUsed := p_PdnTypeToBeUsed;

  }


f_EUTRA_MobileInfo_SetAssignedPdnType
	  function f_EUTRA_MobileInfo_SetAssignedPdnType(PDN_Index_Type p_PdnIndex,

                                                 B3_Type p_PdnType := '000'B) runs on EUTRA_PTC

  { 
    var integer v_Index := f_PDN_PdnIndex2Integer(p_PdnIndex);
    vc_EUTRA_Global.MobileInfo.PdnInfo.AssignedPdnType[v_Index] := p_PdnType;

  }

	


f_EUTRA_MobileInfo_GetPdnTypeToBeUsed
	  function f_EUTRA_MobileInfo_GetPdnTypeToBeUsed() runs on EUTRA_PTC return PDN_TypeToBeUsed_Type

  {

    return vc_EUTRA_Global.MobileInfo.PdnInfo.PdnTypeToBeUsed;

  }


f_EUTRA_MobileInfo_GetAssignedPdnType
	  function f_EUTRA_MobileInfo_GetAssignedPdnType(PDN_Index_Type p_PdnIndex := PDN_1) runs on EUTRA_PTC return B3_Type

  { 
    var integer v_Index := f_PDN_PdnIndex2Integer(p_PdnIndex);
    return vc_EUTRA_Global.MobileInfo.PdnInfo.AssignedPdnType[v_Index];

  }

	  function f_EUTRA_Get_508_TrafficFlowTemplate(integer p_BearerContext,

                                               EPS_BearerIdentity p_EPS_BearerId := tsc_EpsDedicatedBearerId,

                                               template (omit) B3_Type p_PdnTypeValue := omit,

                                               PDN_Index_Type p_PdnIndex := PDN_1) runs on EUTRA_PTC return template (value) TrafficFlowTemplate

  {

    var B3_Type v_PdnTypeValue;

    if (isvalue(p_PdnTypeValue)) {

      v_PdnTypeValue := valueof(p_PdnTypeValue);

    } else {

      v_PdnTypeValue := f_EUTRA_MobileInfo_GetAssignedPdnType(p_PdnIndex);

    }

    return f_EUTRA_NB_Get_508_TrafficFlowTemplate(p_BearerContext, p_EPS_BearerId, v_PdnTypeValue, p_PdnIndex);

  }


f_EUTRA_GetPDNAddress
	  function f_EUTRA_GetPDNAddress(boolean p_IPv4AllocationViaNas,

                                 PDN_Index_Type p_PdnIndex := PDN_1
) runs on EUTRA_PTC return PDN_Address

  { 
    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed := f_EUTRA_MobileInfo_GetPdnTypeToBeUsed();
    var PDN_Address v_PDN_Address := f_EUTRA_NB_GetPDNAddress(p_IPv4AllocationViaNas, p_PdnIndex, v_PDN_TypeToBeUsed);

    var B3_Type v_TypeValue := v_PDN_Address.typeValue;

    f_EUTRA_MobileInfo_SetAssignedPdnType(p_PdnIndex, v_TypeValue);

    return v_PDN_Address;

  }


f_EUTRA_InitialRegistration_Step14_15
	  function f_EUTRA_InitialRegistration_Step14_15(EUTRA_CellId_Type p_CellId,

                                                 EUTRA_ATTACH_Type p_ForcedAttach,

                                                 template (omit) AdditionalUpdateType p_AdditionalUpdateType,

                                                 ProcedureTransactionIdentifier p_EPS_TI,

                                                 boolean p_IPv4AllocationViaNasFlag,

                                                 template (omit) ProtocolConfigOptions p_PcoToUE,

                                                 AccessPointName p_APN,

                                                 PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE, // removed in further changes
                                                 template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,

                                                 template (omit) GPRS_Timer p_T3402 := omit,

                                                 template (omit) GPRS_Timer p_T3423 := omit,

                                                 template (omit) PLMN_List p_EquivalentPlmnList := omit,

                                                 integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@

                                                 template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := f_GetNetworkFeatureSupport(), // @sic R5-155988 sic@

                                                 template (omit) EMM_Cause p_EMM_Cause := omit,

                                                 EUTRA_InitialRegistrationPDN_Type p_InitialRegistrationPDN := firstPDN) runs on EUTRA_PTC

  { 
    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed := f_EUTRA_MobileInfo_GetPdnTypeToBeUsed(); 
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit) EPS_NetworkFeatureSupport v_NtwkFeatSupport := p_NtwkFeatSupport;

    var template (omit) EmergNumList v_EmergNumList := omit;

    var template (omit) MobileIdentity v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@

    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(p_InitialRegistrationPDN);                   // @sic R5-144797, R5s150510 sic@

    var EPS_BearerIdentity v_EpsDefaultBearerId := f_EUTRA_InitialRegistrationEpsBearerId(p_InitialRegistrationPDN);  // @sic R5-144797, R5s150510 sic@

    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@

    var template (value) PDN_Address v_PDN_Address;

    var EmergencyNumList v_EmergencyNumList;

    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_1;

    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */

      select (v_PDN_TypeToBeUsed) {

        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed

        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed

        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }

      }

    }

    if (p_NoOfEmergencyNumbers > 0) {

      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_EUTRA_MobileInfo_GetLocalEmergencyNumberList()); /* @sic R5s141315 sic@ */

      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers

      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);

    }

    if (tsc_EUTRA_Registration_PDN1isIMS) {     // @sic R5-144797, R5s141367 sic@

      v_BearerContextNumber := DEF_2;

    }

    v_PDN_Address := f_EUTRA_GetPDNAddress(p_IPv4AllocationViaNasFlag, v_PdnIndex, p_PDN_TypeToBeUsed); // removed in further changes
    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,

                                                                  v_EpsDefaultBearerId,    // @sic R5-144797 sic@

                                                                  tsc_RRC_TI_Def,

                                                                  p_EPS_TI,

                                                                  v_AttachType,

                                                                  v_PDN_Address,

                                                                  v_Guti,

                                                                  cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  v_ESM_Cause, // @sic R5-104703 sic@

                                                                  p_APN,

                                                                  p_PcoToUE,

                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                  p_T3412,

                                                                  p_T3402,

                                                                  p_T3423,

                                                                  p_EquivalentPlmnList,

                                                                  v_EmergNumList,         // @sic R5s130681 sic@

                                                                  v_NtwkFeatSupport,      // @sic R5s130681 sic@

                                                                  -,

                                                                  -,

                                                                  v_BearerContextNumber, // @sic R5-144797 sic@

                                                                  p_EMM_Cause); // @sic R5-145971 sic@

  }


fl_TC_10_3_1_TestProcedure
	  function fl_TC_10_3_1_TestProcedure(boolean p_IpTypeIsIPv4) runs on EUTRA_PTC

  {

    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed;

    if (p_IpTypeIsIPv4) { v_PDN_TypeToBeUsed := forceIPv4only; }   // @sic R5-120541 sic@

    else                { v_PDN_TypeToBeUsed := forceIPv6only; }   // @sic R5-120541 sic@

    f_EUTRA_MobileInfo_SetPdnTypeToBeUsed(v_PDN_TypeToBeUsed);
    //@siclog "Step 0" siclog@

    // The SS performs the generic procedure to get UE in Loopback Activated (state 4)

    // with a Reference default EPS bearer context #1

    // and a Reference dedicated EPS bearer context #1 according to TS 36.508 on Cell A

    // Now continue with the procedure

    f_EUTRA_InitialRegistration(eutra_CellA, PREAMBLE, STATE2A_TESTLOOP_ModeB, NORMAL); // @sic R5-160947 sic@

    // UE is in Loopback Activated(state 4) according to TS 36.508 on Cell A.

    f_EUTRA_RbEst_508_AM ( eutra_CellA );

    f_EUTRA_CloseUE_TestLoopModeB ( eutra_CellA, '00'O );

    // steps 1-5

    fl_TC_10_3_1_ProcedureSteps1to5(p_IpTypeIsIPv4);

  }


fl_TC_10_9_1_TestProcedure
	  function fl_TC_10_9_1_TestProcedure(boolean p_IpTypeIsIPv4) runs on EUTRA_PTC

  {

    var template (value) EPS_QoSList_Type v_Qos_List := {cs_508_EPS_QoS_Dedicated_1, cs_508_EPS_QoS_Dedicated_1};

    var template (value) Tft_List_Type v_Tft_List := {fl_Get_TFT_10_9_1(p_IpTypeIsIPv4, tsc_EpsDedicatedBearerId),

                                                      fl_Get_TFT_10_9_1(p_IpTypeIsIPv4, tsc_EpsDedicatedBearerId2)};

    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed;

    if (p_IpTypeIsIPv4) { v_PDN_TypeToBeUsed := forceIPv4only; }   // @sic R5-120542 sic@

    else                { v_PDN_TypeToBeUsed := forceIPv6only; }   // @sic R5-120542 sic@

    f_EUTRA_MobileInfo_SetPdnTypeToBeUsed(v_PDN_TypeToBeUsed);
    //@siclog "Step 0" siclog@

    // The SS performs the generic procedure to get UE in Loopback Activated (state 4) on Cell A establishing a default EPS bearer

    // in accordance to Reference default EPS bearer context #1 (DRB1) as specified in subclause 6.6.1 in [18]

    // and two dedicated EPS bearers (DRB2 and DRB3) with EPS bearer context as specified in ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST message

    // for DRB2 and DRB3 in subclause 10.9.1.3.3.

    // Now continue with the procedure

    f_EUTRA_InitialRegistration(eutra_CellA, PREAMBLE, STATE2A_TESTLOOP_ModeB, NORMAL);  // @sic R5-160947 sic@

    // UE is in Loopback Activated(state 4) according to TS 36.508 on Cell A.

    f_EUTRA_GenericRbEst(eutra_CellA, 2, 0, v_Qos_List, v_Tft_List); // TFTs are either IPv4 or IPv6

    f_EUTRA_CloseUE_TestLoopModeB(eutra_CellA, '00'O);

    // steps 1-8

    fl_TC_10_9_1_ProcedureSteps1to8(p_IpTypeIsIPv4);

  }


Further changes
Parameter p_PDN_TypeToBeUsed removed from 
· f_EUTRA_Preamble
· f_EUTRA_InitialRegistration_Step5_18b1
· f_EUTRA_InitialRegistration_Step5_16c1
· f_EUTRA_Preamble_CDMA2000_PreRegistration
· f_EUTRA_InitialRegistration_Step5_15
· f_EUTRA_InitialRegistration_Step14_15
· f_EUTRA_InitialRegistration
